UNIVERSITÉ DE POITIERS **2009-2010**
Agrégation de Mathématiques

Chapitre 5 : corrigés de TP en SCILAB

Exercice 1

```
//Resolution numerique par le schema d'Euler
clear
T=2;
N=10;h=T/N;//pas de discretisation

deff('y=f(t,x)','y=t.*x')
deff('x=sol(t)','x=exp(t.^2/2)') // solution exacte
x0=1;//condition initiale
t0=0;
tabt=[t0];//pour stocker les valeurs de t
tabx=[x0];//pour stocker les valeurs de x
x=x0;
for n=0:N-1
    t=t0+n*h;
    x=x+h*f(t,x);//schema d'Euler x_{n+1}=x_n+hf(tn,xn)
    tabx=[tabx x];
    tabt=[tabt t+h];
end
tt=[0:0.01:T];//pour afficher la solution exacte
 xrk=ode("rk",x0,t0,tt,f);//solution donnee par un solveur Runge-Kutta avec pas adaptatif
clf
plot2d(tt,sol(tt),style=1)
plot2d(tabt,tabx,style=-1)
plot2d(tt,xrk,style=2)
legends(['exacte','solveur scilab','Euler explicite'],[1 2 -1],opt='ul')
xtitle('','t','x')
```

Ce programme produit la figure suivante :

Pour un pas $h = 0.2$, on voit nettement la différence entre la solution exacte et la solution donnée par le schéma d'Euler explicite. Par contre, on ne distingue pas visuellement la différence entre la solution donnée par le solveur (en bleu) et la solution exacte (en noir, sous la courbe en bleu !).

Exercice 2

```
//ordre de la mthode d'Euler
  clear
  clf
  T=2;


  deff('y=f(t,x)','y=t.*x')
  deff('x=sol(t)','x=exp(t.^2/2)') // solution exacte
  taberreur=[];
```



Figure 1: Méthode d'Euler explicite

```
tabN=[];
for k=2:13
    N=2^k;h=T/N; //h = pas de discrtisation
    tabN=[tabN N];
    x=1;//condition initiale
  t0=0;
  tabt=[t0];//pour stocker les valeurs de t
  tabx=[x];//pour stocker les valeurs de x
  for n=0:N-1
      t=t0+n*h;
      x=x+h*f(t,x);//schema d'Euler x_{n+1}=x_n+hf(tn,xn)
      tabx=[tabx x];
      tabt=[tabt t+h];
  end
   erreur=max(abs(tabx-sol(tabt)));//max sur un tableau de xi
   taberreur=[taberreur erreur];

end
clf
plot2d(tabN,taberreur,logflag='ll',style=-1)
plot2d(tabN,10*ones(tabN)./tabN,logflag='ll',style=2) //pour comparer a une droite de pent
legends(['Euler';'O(1/N)'],[-1 2],opt='ll')
xtitle('','N','erreur')
```

Ce programme produit la figure suivante : On vérifie que la méthode d'Euler converge en $O(1/N)$.

Exercice 3

```
//ordre des mthodes d'Euler, du point milieu et de Runge-Kutta
  clear
```



Figure 2: Erreur pour la méthode d'Euler explicite

```
clf
T=2;

deff('y=f(t,x)','y=t.*x')
deff('x=sol(t)','x=exp(t.^2/2)') // solution exacte
taberreurEuler=[];//tableau des erreurs pour Euler explicite
taberreurPM=[];//tableau des erreurs pour le point milieu
taberreurRK=[];//tableau des erreurs pour Runge-Kutta
tabN=[];
for k=2:13
    N=2^k;h=T/N; //h = pas de discrtisation
    tabN=[tabN N];
    t0=0;//temps initial
    x0=1;//condition initiale
    xEuler=x0;
    xPM=x0;
    xRK=x0;
  tabt=[t0];//pour stocker les valeurs de t
  tabxEuler=[xEuler];//pour stocker les valeurs de xEuler
  tabxPM=[xPM];
  tabxRK=[xRK];
  for n=0:N-1
      t=t0+n*h;
      //1. Schema d'Euler
      xEuler=xEuler+h*f(t,xEuler);
      //2. Schema du point milieu
      k1PM=f(t,xPM);
      xPM=xPM+h*f(t+h/2,xPM+h*k1PM/2);
      //3. Schema de Runge-Kutta
```

```
        k1RK=f(t,xRK);
        k2RK=f(t+h/2,xRK+h*k1RK/2);
        k3RK=f(t+h/2,xRK+h*k2RK/2);
        k4RK=f(t+h,xRK+h*k3RK);
        xRK=xRK+h*(k1RK+2*k2RK+2*k3RK+k4RK)/6;
        tabt=[tabt t+h];
        tabxEuler=[tabxEuler xEuler];
        tabxPM=[tabxPM xPM];
        tabxRK=[tabxRK xRK];
    end
     erreurEuler=max(abs(tabxEuler-sol(tabt)));//max sur un tableau de xi
     erreurPM=max(abs(tabxPM-sol(tabt)));
     erreurRK=max(abs(tabxRK-sol(tabt)));

     taberreurEuler=[taberreurEuler erreurEuler];
     taberreurPM=[taberreurPM erreurPM];
     taberreurRK=[taberreurRK erreurRK];

  end
  clf
  plot2d(tabN',[taberreurEuler' taberreurPM' taberreurRK'],logflag='ll',style=[-1 -2 -3])
   legends(['Euler';'Point milieu';'Runge-Kutta']  ,[-1 -2 -3],opt='ll')
  xtitle('','N','erreur')
```

Ce programme produit la figure suivante :



Figure 3: Erreur pour les méthodes d'Euler, du point milieu et de Runge-Kutta

On vérifie que la méthode d'Euler est d'ordre 1, celle du point milieu d'ordre 2 et celle de Runge-Kutta d'ordre 4.